Seminar Programmiersprachen im Verlgeich - Zig

Tobias Schmitz

Contents

1. Introduction

The objective of this seminar was getting to know a new programming language by writing a simple grep [1] like program.

2. Language

2.1. About

Zig is a general-purpose compiled systems programming language. It was initially developed by Andrew Kelley and released in 2015/2016.

- **TODO**
- Zig software foundation
  - only no strings attached donations
- It is often mentioned as a successor to C.

2.2. Toolchain

Installation was as simple as downloading the release tar archive from the downloads section of the Zig website [2], extracting the toolchain, and symlinking the binary onto a `$PATH`. There is also a community project named `zigup` [3] which is allows installing and managing multiple versions of the Zig compiler.

The compiler is a single executable named `zig`, it includes a build system which can be configured in a `build.zig`, which is written in Zig [4]. The toolchain itself is also a C/C++ compiler which allows Zig code to directly interoperate with existing C/C++ code. [5]

To start a new project inside an existing directory running `zig init-exe` will generate the main source file `src/main.zig` and the build configuration `build.zig`. The program can then be built and executed by running `zig build run`. [6]

The Zig community also provides a language server named `zls` [7], which worked right away after setting it up in `neovim` [8]. There were some issues with type inference, and completion of member functions of generic types. In some cases `zls` would report no errors when the Zig compiler would.

2.3. Integers

Zig has concrete integer types with a predetermined size regardless of the target platform. Commonly used types are:

- unsigned: `u8`, `u16`, `u32`, `u64`
- signed: `i8`, `i16`, `i32`, `i64`.

But it also allows defining both signed and unsigned, arbitrary sized integers like `i27`, or `u3`, up to a maximum bit-width of `65535` [9].

2.4. Arrays and slices

Zig arrays [10] have a size known at compile time and are stack allocated, unless specified otherwise:

```
1  const explicit_length = [5]u32{ 0, 1, 2, 3, 4 };
2  const inferred_length = [_]u32{ 5, 6, 7, 8, 9 };
```

Arrays can be sliced using the index operator with an end-exclusive or half-open range, returning a slice [11] to the referenced array. By default the slice is a fat pointer which is composed of a pointer that points to the memory address of the slice inside the array, and the length of the slice.

```
1  const array = [_]u32{ 0, 1, 2, 3, 4 };
2  const slice = array[1..3];
3  std.debug.print("{*}\n{d}\n", .{ slice, slice.* });
```

The code prints the fat pointer itself, and the dereferenced values of the array it points to:

```
[2]u32@2036fc
{ 1, 2 }
```

For better interoperability with C Zig also allows sentinel terminated pointers [12] or slices. Most commonly this is used for null-terminated strings:

```
const string: *const [32:0]u8 = "this is a null terminated string";
const fat_pointer_slice: []const u8 = string[10..];
const null_term_slice: [:0]const u8 = string[10..];
const null_term_ptr: [*:0]const u8 = string[10..];

std.debug.print("size: {}, '{s}'\n", .{@sizeOf([]const u8), fat_pointer_slice});
std.debug.print("size: {}, '{s}'\n", .{@sizeOf([:0]const u8), null_term_slice});
std.debug.print("size: {}, '{s}'\n", .{@sizeOf([*:0]const u8), null_term_ptr});
```

The code prints the size of the pointer and the string it references:

```
size: 16, 'null terminated string'
size: 16, 'null terminated string'
size: 8, 'null terminated string'
```

All three slice/pointer types reference the same data, but in a different way. Variant 1 uses the fat pointer approach described above. Variant 2 uses the same approach but also upholds the constraint that the end of the slice is terminated by a null-byte sentinel. Variant 3 only stores a memory address and relies upon the null-byte sentinel to compute the length of the referenced data when needed.
On 64-bit target platforms the first and the second slice type have a size of 16 bytes, 8 bytes for the pointer and 8 additional bytes for the length. The sentinel terminated pointer only has a size of 8 bytes, since it doesn't store an additional length field.
A limitation of sentinel terminated slices or pointers is that they cannot reference arbitrary parts of an array. Trying to do so fails with the following message:

```
src/main.zig:10:62: error: expected type '[:0]const u8', found '*const [13]u8'
    const null_terminated_string_slice: [:0]const u8 = string[10..23];
                                                        ~~~~~~^~~~~~~~
src/main.zig:10:62: note: destination pointer requires '0' sentinel
```

2.5. Type inference

The type of Zig variables is inferred using only directly assigned values. A type can optionally be specified, and is required in some cases. For example when an integer literal is assigned to a mutable variable, it's exact type must be specified. When the type of a `struct` is known, such as when passing it to a function, it's name can be omitted and an anonymous literal can be used:

```
const Foo = struct {
    a: i32,
    b: bool = false,
};
fn doSomething(value: Foo) void { ... }

doSomething(.{ .a = 21 });
```

The same is also true for an `enum` and a tagged `union`. When the type is known, the name of the enum can be omitted and only the variant needs to written out:

```
const Bar = enum {
    One,
    Two,
    Three,
};
const BAR_THREE: Bar = .Three;

const Value = union(enum) {
    Int: u32,
    Float: f32,
    Bool: bool,
};
const INT_VAL: Value = .{ .Int = 324 };
```

2.6. Tagged unions

In Zig tagged unions are very similar to Rust enums [13]. The tag can be either an existing enum or inferred from the union definition itself. If an existing enum is used as a tag, the compiler enforces that every variant that the enum defines is present in the union declaration. Tagged unions can be coerced to their enum tag, and an enum tag can be coerced to a tagged union when it is known at `comptime` and the union variant type has only one possible value such as `void`. [14]

```
const TokenType = enum {
    Ident,
    IntLiteral,
    Dot,
};
const Token = union(TokenType) {
    Ident: []const u8,
    IntLiteral: u64,
    Dot,
};

const ident_token = Token { .Ident = "abc" };
const token_type: TokenType = ident_token;
// `Token.Dot` has only one value
const dot_token: Token = TokenType.Dot;
```

2.7. Control flow

Zig has special control flow constructs for dealing with union and optional values. Optional values can be passed into an if statement and if a non-null value is contained, it can be accessed inside the if statement's body:

```
const optional_int: ?u32 = 86;
if (optional_int) |int| {
    std.debug.print("Contains int {}\n", .{int});
}
```

To unwrap optional values more conveniently zig provides the `orelse` operator which allows specifying a default value, and the `.?` operator which forces a value to be unwrapped and will crash if it is `null`.

```
const optional_token: ?Token = parser.next();
const token = optional_token orelse unreachable;
// syntax sugar for the above
const token = optional_token.?;
```

Switch statements can be used to extract the values of tagged unions in a similar way:

```
switch (token) {
    .Ident => |ident| {
        std.debug.print("Identifier: '{}'", .{ident});
    }
    .IntLiteral => |int| {
        std.debug.print("Integer literal: '{}'", .{int});
    }
    .Dot => {},
}
```

2.8. Error handling

In Zig there are no exceptions and errors are treated as values. If a function is fallible it returns an error union, the error set of that union can either be inferred or explicitly defined.

```
// inferred error set
pub fn main() !void {
    const num = try std.fmt.parseInt(u32, "324", 10);
    //          ^ unwraps or returns the error
    std.debug.print("{d}\n", .{num});
}

// named error set
const IntError = error{
    IsNull,
    Invalid,
};
fn checkNull(num: usize) IntError!void {
    if (num == 0) {
        return error.IsNull;
    }
}
```

Like optional values, error unions can be inspected using an if statement. The only difference is that the else clause now also grants access to the error value:

```
fn fallibleFunction() !f64 { ... }

if (fallibleFunction()) |val| {
    std.debug.print("Found value: {}\n", .{val});
} else |err| {
    std.debug.print("Found error: {}\n", .{err});
}
```

And likewise the `catch` operator for error unions corresponds to the `orelse` operator of optionals.

```
const DEFAULT_PATH = "$HOME/.config/zig-grep";
const path = readEnvPath() catch DEFAULT_PATH;
```

Additionally the `catch` operator allows capturing the error value.
Similar to Rust [15] Zig has a `try` operator that either returns the error of an error union from the current function or unwraps the value:

```zig
const file = openFile() catch |err| return err;
// syntax sugar for the above
const file = try openFile();
```

2.9. Defer

There are no destructors in Zig, so unlike C++ where the RAII [16] model is often used to make objects manage their resources automatically, resources have to be managed manually. A commonly used pattern to manage resources is for an object to define `init` and a `deinit` procedures, which have to be called manually.

To make this more ergonomic Zig provides `defer` statements, which allow running cleanup code when a value goes out of scope. If multiple `defer` statements are defined, they are run in reverse declaration order. This allows the deinitialization code to be directly below the initialization: [17]

```zig
fn fallibeFunction() ![]const u8 {
    var foo = try std.fs.cwd().openFile("foo.txt", .{});
    defer foo.close();
    var bar = try std.fs.cwd().openFile("bar.txt", .{});
    defer bar.close();

    // the defer statements will be executed in this order:
    // 1. bar.close();
    // 2. foo.close();
}
```

The `errdefer` statement runs code **only** when an error is returned from the scope, this can be useful when dealing with a multi step initialization process that can fail, and intermediate resources need to be cleaned up: [18]

```zig
fn openAndPrepareFile() !File {
    var file = try std.fs.cwd().openFile("foo.txt", .{});
    errdefer file.close();
    try file.seekBy(8);
    return file;
}
```

If the function succeeds the file is returned from it, so it shouldn't be closed. If it fails while seeking and returns an error, the `errdefer` statement is executed and the file is closed as to not leak any resources.

2.10. Memory management

Zig doesn't include a garbage collector and uses a very explicit manual memory management strategy. Memory is manually allocated and deallocated via `Allocator`s that are choosen and instantiated by the user. Data structures or functions which might allocate require an allocator to be passed explicitly. The standard library includes a range of allocators fit for different use cases, ranging from general purpose bucket allocators, bump- or arena allocators, to fixed buffer allocators.
Memory allocation may fail, and out of memory errors must be handled. Memory deallocation must always succeed. Like other resources, data structures that allocate commonly provide an `init` and `deinit` procedure which can be used in combination with a `defer` statement to make sure allocated memory is freed.
In debug mode the `GeneralPurposeAllocator` keeps track of allocations and detects memory leaks and double frees [19].

2.11. Comptime

Zig provides a powerful compile time evaluation mechanism to avoid using macros or code generation. Contrary to C++ or Rust where functions have to be declared `constexpr` [20] or `const` [21]

in order to be called at compile time, in Zig everything that can be evaluated at `comptime` just is. A function called from a `comptime` context will either yield an error explaining what part of it isn't able to be evaluated at `comptime` or evaluate the value during compilation. A `comptime` context can be a constant defined at the global scope, or a `comptime` block. [22]

This can be used to do expensive calculations, generate lookup tables, or uphold constraints using assertions, all at compile time:

```
const FIB_8 = fibonacci(8);
comptime {
    // won't compile
    std.debug.assert(fibonacci(3) == 1);
}

fn fibonacci(n: usize) u64 {
    if (n == 0 or n == 1) {
        return 1;
    }
    var a = 1;
    var b = 1;
    for (1..n) |_| {
        const c = a + b;
        a = b;
        b = c;
    }
    return b;
}
```

This means, if the main function is able to be evaluated at compile time, and it is called from a `comptime` context, the Zig compiler acts as an interpreter and the program is executed during compilation. Granted since comptime code can't perform I/O such a program is quite limited.

Arguments to functions can be declared as `comptime` which requires them to be known during compilation. This is often used for types passed to function in the same way other languages handle generics. Considering the following Kotlin [23] class:

```
class Container<T>(
    var items: ArrayList<T>,
)
```

An equivalent Zig `struct` would be defined as a function taking a `comptime` type as an argument that returns another type.

```
fn Container(comptime T: type) type {
    return struct {
        items: ArrayList(T),
    }
}
```

Note that ArrayList is another such function defined in the std library.

2.12. SIMD

In addition to the automatic vectorization of code that the LLVM [24] optimizer does, Zig also provides a way to explicitly define vector operations, using the `@Vector` intrinsic, that will compile down to target specific SIMD operations: [25]

```
const a = @Vector(4, i32){ 1, 2, 3, 4 };
const b = @Vector(4, i32){ 5, 6, 7, 8 };
const c = a + b;
```

2.13. Ecosystem

Compared to C++, Java, or Rust the std library of Zig is quite minimal. Neither the Zig language itself, nor the std library directly define a string datatype. String literals are represented as byte slices (`[]const u8`), which allows using the whole range of `std.mem.*` functions to operate on them.

- somewhat immature ecosystem
  - missing regex library
  - async not available in `0.11` self-hosted compiler

3. Development process

Since the scope of the program was predetermined, I mainly focused on performance.

The Zig std library provides `IterableDir`, an iterator for iterating a directory in a depth first manner, but unfortunately that approach doesn't allow filtering of searched directories. To overcome that limitation I mostly copied the std library function for iterating directories and modified it slightly to allow filtering out hidden directories.

There are some beginnings of regex libraries written in Zig, but they are still in their infancy, and aren't feature complete. So I decided on using the Rust regex library (`rure`) [26] through it's C API, since it's a standalone project without tethers to a standard library, and reasonably fast [27].
To include a C library, some modification inside the `build.zig` configuration file are needed:

```
// link all the other stuff needed
exe.linkLibC();
exe.linkSystemLibrary("util");
exe.linkSystemLibrary("dl");
exe.linkSystemLibrary("gcc_s");
exe.linkSystemLibrary("m");
exe.linkSystemLibrary("rt");
exe.linkSystemLibrary("util");
// link rure itself
exe.addIncludePath(LazyPath.relative("rure/regex-capi/include"));
exe.addLibraryPath(LazyPath.relative("rure/target/release"));
exe.linkSystemLibrary2("rure", .{
    .needed = true,
    .preferred_link_mode = .Static,
});
```

This links the `rure` crate statically into the final binary.

The dependencies are taken straight from the `rure` compile script:

```
# N.B. Add `--release` flag to `cargo build` to make the example run faster.
cargo build --manifest-path ../Cargo.toml
gcc -O3 -DDEBUG -o iter iter.c -ansi -Wall -I../include -L../../target/debug -
lrure
# If you're using librure.a, then you'll need to link other stuff:
# -lutil -ldl -lpthread -lgcc_s -lc -lm -lrt -lutil -lrure
```

When linking C libraries, Zig isn't able to include debug symbols, so crash messages that would normally be informative, only show memory addresses:

```
thread 20843 panic: index out of bounds: index 14958, len 14948
Unwind error at address `:0x2ebaef` (error.InvalidDebugInfo), trace may be
incomplete
```

This is a known issue [28].

The C functions can then be imported using the `@cImport` intrinsic:

```
const c = @cImport({
    @cInclude("rure.h");
});
```

And the c definitions can be accessed using the returned object.

```
var match: c.rure_match = undefined;
const found = c.rure_find(ctx.regex, @ptrCast(text), text.len, pos, &match);
```

3.1. Single threaded optimization

3.1.1. Line by line searching

To keep it simple the first implementation, read the whole file into a single buffer and ran a compiled regex pattern match on every line. This was done using a regex iterator from the `rure` crate.

3.1.2. Whole text searching

After some investigation it turned out that initializing the regex search iterator provided by the `rure` crate had more overhead than expected, and running the regex pattern match on the whole text instead of every line would improve performance significantly. Following the previous change, I found out that the `rure` library provided a function that allowed setting the start index for searching inside the passed text slice. Using this function avoided allocating the iterator in the first place.

3.1.3. Line by line searching with fixed size buffer

Since one of the tests was to search an 8gb large text file, the input would need to be split up into smaller chunks as to avoid running out of memory. This was done using a fixed size buffer which would only load part of the file, searching that buffer up to the last fully included line, then moving the unsearched parts including possibly relevant context lines to the start of the buffer, and eventually refilling the buffer with remaining data to search. Since lines need to be iterated to calculate line numbers, and I discovered the `rure` function that searches the text directly without an iterator, I decided to once again search each line individually, instead of the whole text.

3.1.4. Whole text searching with fixed size buffer

After further investigation I discovered that the overhead of searching each line didn't just come from the `rure` iterator, but that special regex patterns introduced large overhead when starting the search. One example was the word character pattern `\w`, which respects possibly multi-byte unicode characters. Since the `rure` library uses a finite automata (state machine), matching multiple word characters results in a large number of states [29]. This state machine, although only compiled once, needs to be initialized in memory every time a search is starts. Disabling unicode support during the regex pattern compilation significantly improved performance. With these findings, the regex pattern matching was once again adjusted to be run on the whole text buffer, to restore previously achieved performance.
One additional bug that I only tackled at this stage was to prevent regex pattern matches that spanned multiple lines. If a match is found that spans multiple lines an additional search is run only on the fist matched line, if this succeeds too only this match is highlighted and printed.

3.2. Parallelization

At this point most easy wins in single threaded optimization were off the table, so the next performance improvements would come from using multiple threads. The most time consuming sections of the program are accessing the file system, and searching the text.

Parallelization was implemented using a thread pool of search workers that would receive file paths using a atomically synchronized message queue. The directory walking remained mostly the same apart from searching files adhoc, they were now sent through the message queue.

The output of multiple threads now had to be synchronized so that lines printed from one file would not be interspersed with other ones.
There are two obvious solutions to this problem. One is to use a dynamically growing allocated buffer which stores the entire output of a searched file and then write the entire buffer in a synchronized way when the file is fully searched. This would avoid blocking other threads, but could cause the program to run out of memory if large portions of big files would match a search pattern.
The other solution is to just block output of all other threads once a match has been found in a file and then write all lines directly to stdout. This would avoid running out of memory, but could in worst case scenarios cause basically single threaded performance.
The final implementation uses a hybrid of the two, each thread has a fixed size output buffer which can be written to without any synchronization. Once the buffer is full access to stdout is locked for other threads until the file is fully searched, but other workers can still access their thread-local output buffers.

Text searching had been parallelized the search workers were now emptying the message queue too quickly, so walking the file system with multiple threads was up next.
This was heavily influenced by the Rust [30] `ignore` by the same author as, and also used in ripgrep [31]. A thread pool of "walkers" is used to search multiple directories simultaneously in a depth first manner to reduce memory consumption. A walker tries to pop of a directory iterator of a shared atomically synchronized stack, by blocking until one is available. Once it receives a directory it iterates through the remaining entries enqueueing any files encountered. If it encounters a subdirectory, the parent directory is pushed back onto the stack and the subdirectory is walked. Once all walkers are waiting for a new directory iterator all directories have been walked completely and the thread pool is stopped.

3.3. Command line argument parsing

Argument parsing makes use of tagged unions and `comptime`.

There are two different types of arguments: flags and values, both of these are defined as `enum`s. `UserArgFlag`s don't require a value and are just boolean toggles. `UserArgValue`s require a value, for example a number, to be specified after them. `UserArgKind` is a tagged union that either contains one or the other.

```
const UserArgKind = union(enum) {
    value: UserArgValue,
    flag: UserArgFlag,
};

const UserArgValue = enum(u8) {
    Context,
    AfterContext,
    BeforeContext,
};
const UserArgFlag = enum(u8) {
    Hidden,
    FollowLinks,
    Color,
```

```
    NoHeading,
    IgnoreCase,
    Debug,
    NoUnicode,
    Help,
};
```

All user args are defined in an array, including their long form, an optional short form, a description and their union representation.

```
const USER_ARGS = [_]UserArg{
    .{
        .short = 'A',
        .long = "after-context",
        .kind = .{ .value = .AfterContext },
        .help = "prints the given number of following lines for each match",
    },
    ...
    .{
        .short = null,
        .long = "help",
        .kind = .{ .flag = .Help },
        .help = "print this message",
    },
    ...
};
```

When parsing command line arguments this can be used to exhaustively match all possible valid inputs using a switch statement. When adding a new `enum` variant the compiler enforces it is handled in all switch statements that match the modified `enum`. This is the simplified switch statements that handles all arguments:

```
switch (user_arg.kind) {
    .value => |kind| {
        ...

        switch (kind) {
            .Context => {
                opts.after_context = num;
                opts.before_context = num;
            },
            ...
        }
    },
    .flag => |kind| {
        ...

        switch (kind) {
            .Hidden => opts.hidden = true,
            ...
        }
    },
}
```

The help message is generated at `comptime`, using the list of possible arguments.
Instead of a general purpose allocator a fixed buffer allocator had to be used, but otherwise the code could be written without taking any precautions.

3.4. Compiler bug

With Zig `0.11.0` I encountered a bug in the compiler which would affect command line argument parsing. In debug mode arguments were parsed fine, bug in release mode the `--ignore-case` flag would be parsed as the `--hidden` flag. All flags are defined as an `enum`:

```
const UserArgFlag = enum {
    Hidden,
    FollowLinks,
    Color,
    NoHeading,
    IgnoreCase,
    Debug,
    NoUnicode,
    Help,
};
```

The issue was fixed by specifying a concrete tag type to represent the `enum` instead of letting the compiler infer the type.

```
@@ -27,12 +27,12 @@ const UserArgKind = union(enum) {
     flag: UserArgFlag,
 };

-const UserArgValue = enum {
+const UserArgValue = enum(u8) {
     Context,
     AfterContext,
     BeforeContext,
 };
-const UserArgFlag = enum {
+const UserArgFlag = enum(u8) {
     Hidden,
     FollowLinks,
     Color,
```

At the time I discovered the bug, it was already fixed on the Zig `master` branch.

4. Conclusion

5. Bibliography

[1] Jan. 03, 2024. [Online]. Available: https://www.gnu.org/software/grep/manual/grep.html

[2] “Releases”, Zig. [Online]. Available: https://ziglang.org/download/

[3] “zigup”. Jan. 03, 2024. [Online]. Available: https://github.com/marler8997/zigup

[4] “Zig Build System”, Zig. [Online]. Available: https://ziglang.org/learn/build-system/

[5] “Zig”. [Online]. Available: https://ziglang.org/

[6] “Getting Started”, Zig. [Online]. Available: https://ziglang.org/learn/getting-started/

[7] “zls”. Jan. 03, 2024. [Online]. Available: https://github.com/zigtools/zls

[8] “neovim”. [Online]. Available: https://neovim.io/

[9] “Integers”. Jan. 03, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#Integers

[10] “Arrays”. Jan. 03, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#Arrays

[11] “Slices”. Jan. 03, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#Slices

[12] “Pointers”. Jan. 03, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#Pointers

[13] “Enums and Pattern Matching”. [Online]. Available: https://doc.rust-lang.org/book/ch06-00-enums.html

[14] “Tagged union”. Jan. 03, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#Tagged-union

[15] “A Shortcut for Propagating Errors: the ? Operator”. Jan. 04, 2024. [Online]. Available: https://doc.rust-lang.org/book/ch09-02-recoverable-errors-with-result.html#a-shortcut-for-propagating-errors-the--operator

[16] “RAII”. Jan. 04, 2024. [Online]. Available: https://en.cppreference.com/w/cpp/language/raii

[17] “defer”. Jan. 04, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#defer

[18] “errdefer”. Jan. 04, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#errdefer

[19] “GeneralPurposeAllocator source code”. Jan. 04, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/std/src/std/heap/general_purpose_allocator.zig.html

[20] “constexpr specifier”. Jan. 04, 2024. [Online]. Available: https://en.cppreference.com/w/cpp/language/constexpr

[21] “Constant evaluation”. Jan. 04, 2024. [Online]. Available: https://doc.rust-lang.org/reference/const_eval.html

[22] “comptime”. Jan. 04, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#comptime

[23] “Kotlin”. [Online]. Available: https://kotlinlang.org/

[24] “LLVM”. [Online]. Available: https://llvm.org/

[25] “Vectors”. Jan. 04, 2024. [Online]. Available: https://ziglang.org/documentation/0.11.0/#Vectors

[26] “rust-lang/regex”. Jan. 03, 2024. [Online]. Available: https://github.com/rust-lang/regex

[27] Andrew Gallant (BurntSushi), “rebar”. Jan. 03, 2024. [Online]. Available: https://github.com/BurntSushi/rebar

[28] “Invalid debug info when linking to system library.”. [Online]. Available: https://github.com/ziglang/zig/issues/12046

[29] “Simple regex like \w256 take tens of milliseconds to compile”. [Online]. Available: https://github.com/rust-lang/regex/issues/1095

[30] “Rust”. [Online]. Available: https://www.rust-lang.org/

[31] “ripgrep”. Jan. 03, 2024. [Online]. Available: https://github.com/BurntSushi/ripgrep/